
지금까지본코드예제는매우선형적(linear)이다. 즉 모든선언문이맨앞에있

는 것부터 맨 뒤에 있는 것까지 순서대로 실행된다. 선형 코드를 사용하면 결과가

언제나 똑같다. 이와는 대조적으로 조건문은 주어진 조건이 성립될 때만 어떤 작업

을 처리하는 유형의 선언문이다. 선형 코드에서는 인터프리터에서 선언문 A, 선언

문 B, 선언문 C를순서대로실행한다. 조건문을사용하면인터프리터에서선언문 A

를실행하고나서조건 X에따라선언문 B 또는선언문 C 중하나를실행하도록만

들수있다.

조건문을 이용하면 두 개 이상의 결과가 나올 수 있는 상황을 만들고 제어할 수

있다. 예를들어비밀번호로보호된사이트를만드는경우를생각해보자. 사용자가

로그인했을 때 비밀번호가 맞는 경우에는 사용자를 사이트로 들여보내고, 그렇지

않으면사용자에게오류메시지를내보낸다. 이와같이두가지결과를만들려면무

비의스크립트에서로다른코드블록을집어넣어야한다. 한 블록에서는플래시플

레이헤드를 사이트 시작 화면으로 보내야 하는 반면, 다른 한 블록에서는 플레이헤

드를 오류 메시지를 출력하는 프레임으로 보내야 한다. 하지만 사용자가 로그인할

때는 둘 중 한 코드 블록만 실행된다. 조건문을 이용하면 상황에 따라 적절한 블록

을실행하고나머지블록은건너뛰도록만들수있다.

인터프리터에서는 실행할 코드 블록을 어떻게 결정할까? 조건문을 정의할 때는

첫째코드블록을실행시키기위해필요한조건을지정한다. 이 조건이만족되지않

으면 다른 코드 블록이 실행된다(또한 그 코드 블록에도 별도의 조건이 있을 수 있

다). 조건문을 사용할 때는 결국 아래에 유사코드로 표시한 것과 비슷한 구조를 가

진플로우차트와유사한논리구조를만들어야한다.

if (첫 번째 조건) {

// 이 코드를 실행한다.

} else if (두 번째 조건) {

// 이 코드를 실행한다.

} ...그렇지 않으면 {

// Execute this code

}

물론 각 조건은 인터프리터에서 이해할 수 있는 방법으로 기술해야 한다. 하지

만 너무 어렵게 생각할 필요는 없다(지금부터 배우는 내용만 제대로 이해하면 된다).

개념적으로볼때모든조건문은주어진조건을바탕으로코드블록을실행할지아

니면실행하지않을지를결정하는역할을한다.

if 선언문
if 선언문은 가장 일상적인 만능 조건문이다. if 선언문은 갈림길과 비슷하게 코

드의 진행 방향을 두 갈래로 나누기 위해 사용한다. if 선언문에는 한 개 이상의 하

위 선언문이 있어서 주어진 조건이 성립되면 그 선언문을 실행하도록 되어 있다. if

선언문의사용법은다음과같다.

if (condition) {

substatements

}

모든 if 선언문은 if라는키워드로시작한다. 괄호 안에있는주어진조건이만족

되어야하위선언문이실행된다. substatements는하나이상의액션스크립트선언문

이다. 각 하위 선언문은 서로 다른 줄에 있어야 하며 세미콜론으로 끝나야 한다. if

선언문전체는오른쪽중괄호(})로끝나며그뒤에는세미콜론이붙지않는다.

if 선언문의 condition 부분에는 제대로 된 표현식이라면 어떤 표현식이라도 사

용할 수 있다. if 선언문을 실행하면 인터프리터에서는 주어진 표현식(기준 표현식,

test expression)의 값을 확인한다. 기준 표현식 값이 true이면 substatements가 실

행되고 false이면실행되지않는다. 아래 예에서는단순한부울값을기준표현식으

로사용한다.

if (true) {

trace("The condition was met!");  // 이 선언문은 실행된다.

}

if (false) {

trace("The condition was met!");  // 이 선언문은 절대 실행되지 않는다.

}

물론 부울 리터럴은 값이 고정되어 있기 때문에 실제로 기준 표현식에 부울 리

터럴을사용하는경우는없다. 대신부울값을리턴하는복합표현식을사용하게된

다. 예를 들어 비교 연산과 관련된 표현식은 부울 값을 리턴하므로 조건문의 기준

표현식으로쓰기에적합하다.

var pointerX = _xmouse;  // 마우스의 수평 위치

// pointerX > 300 이 true이면
if (pointerX > 300) {

// 이 선언문을 실행한다.

trace("The mouse is past the 300 pixel mark");

}

이제 조금 더 재미있는 부분을 살펴보자. 조건문의 기준 표현식이 반드시 부울

값이어야만하는것은아니므로아무표현식이나그자리에들어갈수있다. 다음과

같이문자열이나숫자를조건문의기준표현식으로사용할수도있다.

if ("hi") {

trace("The condition was met!");

}

if (4) {

trace("The condition was met!");

}

“hi”와 4는 부울형이 아닌데 어떻게 위와 같은 코드가 동작할 수 있을까? 그 해

답은 [표 3-3]에 나온데이터형변환에서찾을수있다. 조건문의기준표현식이부

울 값이 아닌 경우에는 인터프리터에서 그 표현식을 부울형으로 변환한다. 예를 들

어 인터프리터에서는“hi”를 false로 변환한다. 숫자가 아닌 문자열은 부울 값으로

따지면 false가 되기 때문이다. 따라서 이 조건은 성립되지 않기 때문에 위에 있는

첫째 trace() 선언문은 실행되지 않는다. 또한 인터프리터에서는 숫자 4를 true로

변환하므로(0이아닌모든숫자는 true가된다) 둘째 trace() 선언문은실행된다.

드디어 데이터형 변환에 대해 열심히 배운 효과가 나타나고 있다. 아래 예는 이

를응용한몇가지예제이다. 각선언문의실행여부를맞춰보자.

x = 3;

if (x) {

trace("x is not zero");

}

아래예제에서는‘5장. 연산자’에서배운 OR 연산자를이용한다.

lastName = "";

firstName = "";

if (firstName != "" || lastName != "") {

trace("Welcome" + firstName + " " + lastName);

}

마지막으로무비클립객체가존재하는지확인하는예제이다.

if (myClip) {

myClip._x = 0;  // myClip이 존재하면 그 클립을
// 스테이지의 왼쪽 끝에 붙인다.

}

else 선언문
if 선언문만 사용해도 조건에 따라 하나의 코드 블록을 실행시키는 것을 처리할

수있다. 여기에 else 절을사용하면두개의코드블록중어떤것을실행할지결정

할수있다. else 선언문은다음과같이 if 선언문을확장시킨형태로쓰인다.

if (condition) {

substatements1

} else {

substatements2

}

제대로 된 표현식이라면 모두 condition 자리에 쓰일 수 있다. condition이 true

이면 statements1이, false이면 statements2가 실행된다. 즉 else 선언문은 서로 배

타적인결정을내려야할때쓰인다. 둘중한코드블록만실행되기때문이다.

아래코드를살펴보자.

var lastName = "Grossman";

var gender = "male";

if (gender == "male") {

trace("Good morning, Mr." + lastName + ".");

} else {

trace("Good morning, Ms." + lastName + ".");

}

else 절은 종종 if 선언문을 뒷받침하기 위한 용도로 쓰인다. 비밀번호를 입력해

야 들어갈 수 있는 웹사이트 예제를 다시 떠올려보자. 비밀번호가 맞으면 사용자가

사이트에들어갈수있지만그렇지않으면오류메시지만화면에나타나게된다. 패

스워드를조사하는코드는다음과같은형태로쓸수있다(userName과 password는

사용자가입력한정보이고 validUser와 correctPassword는사이트에들어갈수있는로

그인값이라고가정하자).

if (userName == validUser && password == correctPassword) {

gotoAndPlay("intro");

} else {

gotoAndStop("loginError");

}

조건 연산자

두부분으로이루어지는간단한조건문은조건연산자(?:)를 이용하면간편하게

표현할 수 있다. 조건 연산자에는 세 개의 피연산자가 필요하며, if-else 선언문의

각요소와거의같다고보면된다.

condition ? expression1 : expression2;

조건연산자에서는 condition이 true이면 expression1을계산하여그값을리턴

한다. 그렇지 않으면 expression2를 계산하여 그 결과를 리턴한다. 조건 연산자는

다음과같은조건문을간략하게표현한것이라고볼수있다.

if (condition) {

expression1

} else {

expression2

}

조건문과마찬가지로조건연산자를이용하면프로그램의흐름을제어할수있다.

// command가 "go"이면 무비를 재생하고 그렇지 않으면 멈춘다.

command == "go" ? play() : stop();

조건연산자를이용하면변수에값을대입하는것도간단하게표현할수있다.

var guess = "c";

var answer = "d";

var response = (guess == answer) ? "Right": "Wrong";

위코드는아래코드와똑같은기능을한다.

var guess = "c";

var answer = "d";

if  (guess == answer) {

response = "Right";

} else {

response = "Wrong";

}

else if 선언문
if와 else를이용하면조건에따라두개의코드블록중하나를실행할수있다.

if와 else if를사용하면무한히많은코드블록중하나를실행시킬수있다(또는그

중아무것도실행시키지않을수도있다). else와마찬가지로 else if는다음과같이 if

선언문을확장한형태로쓰인다.

if (condition1) {

substatements1

} else if (condition2) {

substatements2

} else if (condition3) {

substatements3

} else {

substatements4  // 위 조건이 모두 맞지 않을 경우에 사용할 선언문
}

condition1, condition2, condition3 자리에는 임의의 유효한 표현식을 사용할

수 있다. condition1이 true이면 substatements1이 실행되고, condition1이 false이

고 condition2가 true이면 substatements2가 실행된다. 그렇지 않으면 codition3을

확인한다. 이런 식으로여러개의 else if 문을이용하여무한히많은조건을검사할

수있다. 기준 표현식중어느것도 true가아니면마지막에있는 else 절에있는선

언문이실행된다. 예를들어로그인검사루틴을다음과같이만들어서다양한오류

메시지를출력할수있다.

if (userName != validUser) {

message = "User not found. Please try again.";

gotoAndStop("loginError");

} else if (password != correctPassword) {

message = "Password incorrect. Please try again.";

gotoAndStop("loginError");

} else {

gotoAndPlay("intro");

}

else if 선언문은사실 else를여러개의 if 선언문과결합시킨것에지나지않는다.

아래에있는두코드는결국같은기능을하지만첫째코드가훨씬이해하기쉽다.

// 일반적인 "else if" 구문
if (x > y) {

trace("x is larger than y");

} else if (x < y) {

trace("x is smaller than y");

} else {

trace("x and y are equal");

}

// if/else를 풀어서 써 놓은 것
if (x > y) {

trace("x is larger than y");

} else {

if (x < y) {

trace("x is smaller than y");

} else {

trace("x and y are equal");

}

}

switch 선언문 흉내내기
액션스크립트에서는 switch 선언문(또는 case 선언문이라고도 부른다)을 지원하

지 않지만 다른 방법으로 비슷한 효과를 낼 수는 있다. switch 선언문을 이용하면

하나의 기준 표현식 값을 기준으로 여러 개의 코드 블록 중 하나만을 실행할 수 있

다. 예를 들면 아래에 나온 자바스크립트의 switch 선언문에서는 기준 표현식인

gender 값에따라각각다른인사말을출력한다.

var surname = "Porter";

var gender = "male";

switch (gender) {

case "femaleMarried": // 기혼 여성의 경우
alert("Hello Mrs. "+ surname);

break;

case "femaleGeneric": // 기혼/미혼을 따지지 않는 경우
alert("Hello Ms. "+ surname);

break;

case "male": // 남성인 경우
alert("Hello Mr. "+ surname);

break;

default :

alert("Hello "+ surname);

}

위 자바스크립트 예제에서는 switch 선언문에서 gender 값을 각 case 표현식에

있는“femealMarried”, “femaleGeneric”, “male”과 비교한다. 여기서는 gender가

“male”이기 때문에 alert(“Hello Mr.”+ surname);이 실행된다. 기준 표현식이 어

떤 조건과도 맞지 않는 경우에는 기본 선언문인 alert(“Hello” + surname);이 실행

된다.

액션스크립트에서 if-else if-else 선언문을 연결해서 사용하면 switch 선언문과

같은효과를연출할수있다.

var surname = "Porter";

var gender = "male";

if (gender == "femaleMarried") {

trace("Hello Mrs. "+ surname);

} else if (gender == "femaleGeneric") {

trace(" Hello Ms." + surname);

} else if (gender == "male") {

trace("Hello Mr."+ surname);

} else {

trace("Hello" + surname);

}

조금 더 고급 기법을 사용한다면 일반적인 객체 속성에 저장된 일련의 함수로

switch를대신할수있다. 이러한방법은 [예제 7-1]에나와있다. 주석을주의깊게

읽어보고이코드가어떻게작동하는지이해하도록하자. 그리고조금전에배운조

건연산자를사용하는방법도자세히살펴보자.

[예제 7-1] switch 선언문 시뮬레이션

var surname = "Porter"; // 사용자의 이름
var gender = "male";    // 사용자의 성(기준 표현식으로 쓰임)

// switch 선언문을 대신할 객체를 만든다.

var mySwitch = new Object();

// mySwitch 객체에 "case expression" 속성을 만든다.

// 각 "case expression" 속성에는 함수가 들어있다.

mySwitch.femaleMarried = function() {

trace("Hello Mrs." + surname);

};

mySwitch.femaleGeneric = function() {

trace("Hello Ms." + surname);

};

mySwitch.male = function() {

trace("Hello Mr." + surname);

};

mySwitch.default = function() {

trace("Hello" + surname);

};

// gender의 값에 따라(이 예제에서는 "male") 적당한 함수를 실행한다.

// 그러한 이름을 가진 속성이 없으면 기본 함수를 실행한다.

mySwitch[gender] ? mySwitch[gender]() : mySwitch["default"]();

간결한 조건문
이책에서는선언문이한줄짜리인경우에도모든조건문에서선언문블록(중괄

호로둘러싸인선언문)을사용했다.

if (x == y) {

trace("x and y are equal");

}

액션스크립트에서는 조건문에 하위 선언문이 하나뿐인 경우에는 중괄호를 사용

하지 않아도 된다. 하위 선언문이 하나뿐인 경우에는 if나 else if 바로 뒤에 중괄호

없이선언문을적어도된다.

if (x == y) trace ("x and y are equal");

또는다음과같이해도된다.

if (x == y)

trace ("x and y are equal");

프로그래머에따라이런스타일을사용하면가독성이떨어지고프로그래밍도중

에실수할가능성이많다고하는사람들도있지만, 이렇게하면소스코드의크기를

줄이는데도움이된다.

앞으로 배울 내용
조건문은액션스크립트에서매우중요한구성요소이다. 조건문을이용하면사용

자가코드의실행조건을제어할수있다. 다음 장에서는또다른중요한실행제어

선언문인순환문에대해알아보도록하자.

